# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-keystoneclient
pkgver=5.2.0
_commit=141787ae8b0db7ac4dffce915e033a78d145d54e
pkgrel=1
pkgdesc="Client Library for OpenStack Identity"
arch=('any')
url="https://docs.openstack.org/python-keystoneclient/latest/"
license=('Apache')
depends=('python-pbr' 'python-debtcollector' 'python-oslo-config' 'python-oslo-i18n'
         'python-oslo-serialization' 'python-oslo-utils' 'python-requests' 'python-six'
         'python-stevedore' 'python-packaging')
makedepends=('git' 'python-setuptools')
checkdepends=('python-fixtures' 'python-keyring' 'python-lxml' 'python-oauthlib'
              'python-os-client-config' 'python-oslotest' 'python-requests-mock'
              'python-stestr' 'python-testresources' 'python-testscenarios' 'python-testtools')
source=("git+https://github.com/openstack/python-keystoneclient.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd python-keystoneclient
  python setup.py build
}

check() {
  cd python-keystoneclient
  stestr run
}

package() {
  cd python-keystoneclient
  python setup.py install --root="$pkgdir" --optimize=1
}
